








Adjustment-Free Inclinometer





This software implements an electrolytic tilt sensor based on the PIC 16C54


(trademark Microchip) processor and MAXIM MAX147 Analog to Digital


Converter.  It contains ADC driver routines, numeric conversions and display


routines that should have utility beyond the example given.  The article is


presented in the Maxim Engineering Journal Volume 29.

; MAX-147 INCLINOMENTER CODE


;


; Maxim Integrated Products


; Applications Engineering


; 120 San Gabriel Drive


; Sunnyvale, CA 94086


;


; Author:


; John M. Wettroth


; Maxim Corporate Apps SE


; 127 Barriedale Circle


; Apex, NC


; (919) 303-5733


;


;********************************************************************


; Demonstration and test code for implementing a electrolytic


; tilt sensor with the Maxim Max-147 2.7V A/D converter and PIC


; 16C54 uP.  System is fully ratiometric and requires no adjustments.


; Displays values on japanese display module for this test version


; of 0 to 4096, with sign.


;********************************************************************


;REVISION HISTORY ; rev0- 3/10/96


; Initial debug- demo operation


; known bugs- handling of -0 and -4095 in two's comp


; clamp values at F00 for this demo version


;********************************************************************





	DEVICE	PIC16C54,XT_OSC,WDT_OFF,PROTECT_OFF





;********************************************************************





; RAM EQUATES- GLOBAL	       ;





ACCaLO	EQU	10h	       ;ACCUM A LOW


ACCaHI	EQU	11h	       ;ACCUM A HI





ACCbLO	EQU	12h	       ;ACCUM B LOW


ACCbHI	EQU	13h	       ;ACCUM B HI





BCDH	EQU	14h	       ;msb- bcd num


BCDM	EQU	15h	       ;			


BCDL	EQU	16h	       ;lsb-





TEMP	EQU	18h	       ;temp byte- temp and dp ctr


DGC	EQU	19H	       ;digit count


DPC	EQU	1AH	       ;image


RBI	EQU	1BH	       ;port b image- used by lcd


; DEBUG


GP1	EQU	1CH	       ;gen purpose reg 1


GP2	EQU	1DH	       ;gen purpose reg 2


CHAR	EQU	1EH	       ;shared- lcd char


COUNT	EQU	1Fh	       ;loop counter- digit counter





; constants/literals


Cy	EQU	0	       ;carry bit


Zr	EQU	2	       ;zero bit


ZSB	EQU	5	       ;pa0 bit - zero suppression


DPT	EQU	2EH	       ;decimal point in ascii





; i/o equates


;LCD


; display data 4-7 are on RB0-3


; EN on RB4


; RS on RB5


EN	EQU	4


RS	equ	5





;sensor on high bits of RB


SEN1	EQU	6


SEN2	EQU	7





;MAX-147 on RA0-3


CS	EQU	0	       ;chip select active low


SCLK	EQU	1	       ;sclk


DIN	EQU	2	       ;din of pic - dout to a/d


DOUT	EQU	3	       ;dout of pic- din to a/d





; ;*******************************************************************


;		     Printf Routine


;  takes bcd number from BCDH, BCDM, BCDL and output ascii number


;  does dgc chars handles leading zero suppression (pa0) and


;  artificial dp placement assumes routine to print a char is


;  say0


;


;  on entry dgc is digit count


;	    dpc is digit to right of dp count (0 for none)


;  examples:   dgc=5 dpc=1 accb= 65535	 6553.6


;	       dgc=5 dpc=0 accb= 65535	 65536


;	       dgc=5 dpc=0 accb= 0	 ____0


;	       dgc=3 dpc=2 accb= 65535	 6.55


;	       dgc=5 dpc=2 accb= 0	 ___.00


;


;


PRF0	MOVLW	05


	MOVWF	DGC


	MOVLW	0


	MOVWF	DPC


PRF	BCF	STATUS,ZSB     ;clear zero suppression bit


	CLRF	GP2	       ;temp storage char cntr


	MOVLW	BCDH	       ;point at high bcd byte


	MOVWF	FSR	       ;point





TOP	MOVF	DPC,0	       ;get dp count


	SUBWF	DGC,0	       ;dgc-dpc if zero then dp now


	BTFSS	STATUS,ZR      ;if zero then cont- else do dp now


	GOTO	NODP	       ;no dp yet


	MOVLW	DPT	       ;load decimal point


	CALL	SAY0	       ;print it


	BSF	STATUS,ZSB     ;set zero supp- right of decimal pt


NODP	MOVF	0,0	       ;get a byte (unswapped)


	BTFSC	GP2,0	       ;if even then low nybble


	SWAPF	0,0	       ;get high nybble


	ANDLW	0FH	       ;mask for only low bits


	BTFSC	STATUS,ZR      ;check for zero


	GOTO	ZP	       ;zero suppression logic


	BSF	STATUS,ZSB     ;non-zero set zsb


ZP	BTFSS	STATUS,ZSB     ;if zero set then do zero


	MOVLW	0F0H	       ;will turn into space when 30h added


	MOVWF	GP1	       ;temp store


	MOVLW	30H	       ;add 30


	ADDWF	GP1,0	       ;add 30 hex- convert to ascii


	CALL	SAY0	       ;print it- either a space or a zero


	INCF	GP2,1	       ;inc char counter


	DECFSZ	DGC,1	       ;dec digit counter


	GOTO	ADJDC	       ;adj digit counter for special case


	RETLW	0	       ;DONE- RETURN





;local "subroutine"- jumps back to top when done


ADJDC	BTFSC	GP2,0	       ;check low bit- new nibble on even


	INC	FSR	       ;next byte


	DECFSZ	DGC,0	       ;see if dig cnt = 1 (no update dgc)


	GOTO	TOP	       ;loop


	BSF	STATUS,ZSB     ;set the zero supp bit- last digit (0)


	GOTO	TOP	       ;loop








;****************************************************************


; MAX-147 SPI ROUTINES


; note that the 147 is s/w compatible with the 186





;CAD   configure and start a/d converter


; w contains value to send to a/d config


; send configuration value from W to a/d and start


CAD	BCF	RA,SCLK        ;just to make sure


	BCF	RA,DOUT        ;


	MOVWF	GP1	       ;stash config byte


	MOVLW	TEMP	       ;point to temp- dummy


	MOVWF	FSR	       ;stash it





	BCF	RA,CS	       ;chip select active


	CALL	SPTR	       ;spi trans and rcv


	RETLW	0	       ;return





;GAD   get a/d routine


; get a/d converter value routine


; enter with w containing pointer


; for high/low will be 1 lower


; retrieve 12 bits from a/d to FSR pointer


; returns a right justified value





GAD	MOVWF	FSR	       ;stash it


	CLRW		       ;zero's to tx


	MOVWF	0	       ;clear fsr- partial fill


	MOVWF	GP1	       ;stash config byte


	MOVLW	5	       ;put 5 bits into high


	CALL	SPTR1	       ;spi trans and rcv- 4 BITS


	DEC	FSR	       ;point to acc low


	CALL	SPTRG	       ;spi trans and rcv-


;fixump


	MOVLW	TEMP	       ;point to temp- dummy


	MOVWF	FSR	       ;stash it


	MOVLW	3	       ;put 4 bits dummy bits in


	CALL	SPTR1	       ;spi trans and rcv- 4 BITS


	BSF	RA,CS	       ;chip select inactive


	RETLW	0	       ;done





;**************


;local subs


;


;SPI support routines


;send 8 bits in W and recieve 8 bits to FSR


;note that chip select must be active prior to this rouine


;W should contain a/d configuration byte with- high bit set


;fsr should point to location where byte to be stored





SPTRG	CLRW		       ;zero's to tx


	MOVWF	GP1	       ;stash config byte


;MAIN ENTRY


SPTR	MOVLW	8	       ;load loop counter 1


SPTR1	MOVWF	COUNT	       ;stash it





MSBLP	BCF	RA,SCLK        ;clock low


	RLF	GP1,1	       ;get high bit to send


	BCF	RA,DOUT        ;preclear dout


	BTFSC	STATUS,CY      ;if in low then skip


	BSF	RA,DOUT        ;set dout


	BSF	RA,SCLK        ;clock high- clock out/get input





	BCF	STATUS,CY      ;preclear carry


	BTFSC	RA,DIN	       ;if in low then skip


	BSF	STATUS,CY      ;set carry


	RLF	0,1	       ;rotate into place- fsr





	DECFSZ	COUNT,1        ;do loop counter


	GOTO	MSBLP	       ;loop for msb


	BCF	RA,SCLK        ;clock low to end


	RETLW	0


;end spi subroutine


;****************************************************************











;****************************************************************


;


;		   Binary To BCD Conversion Routine


;


; standard microchip PIC math toolbox routine- tweaked slightly


;*******************************************************************


;


B2_BCD	BCF	STATUS,0		; clear the carry bit


	MOVLW	16


	MOVWF	COUNT


	CLRF	BCDH


	CLRF	BCDM


	CLRF	BCDL


loop16	RLF	ACCbLO


	RLF	ACCbHI


	RLF	BCDL


	RLF	BCDM


	RLF	BCDH


;


	DECFSZ	count,1


	GOTO	ADJDEC


	RETLW	0


;


ADJDEC	MOVLW	BCDL	       ;should make this loop- save a few bytes


	MOVWF	FSR


	CALL	ADJBCD


;


	MOVLW	BCDM


	MOVWF	FSR


	CALL	ADJBCD


;


	MOVLW	BCDH


	MOVWF	FSR


	CALL	ADJBCD


;


	GOTO	LOOP16


;


ADJBCD	MOVLW	3


	ADDWF	0,W


	MOVWF	TEMP


	BTFSC	TEMP,3		; test if result > 7


	MOVWF	0


	MOVLW	30H


	ADDWF	0,W


	MOVWF	TEMP


	BTFSC	TEMP,7		; test if result > 7


	MOVWF	0		; save as MSD


	RETLW	0





;


;****************************************************************


; LCD DISPLAY SUPPORT


; assumes standard japanese text module


; ( Hitachi 44780 controller )


;


;  3 FUNCTIONS


;   dini- init display


;   tell- send a command to display from acc


;   say0- send data to display from acc  (rem- display not quite ascii)


;   say-  sends data stored in "char" variable (extra entry)





;hardware setup assumed


;    RB4	RS- DATA HIGH/INS LOW


;    RB5	E- ENABLE


;    RB0-3	DATA- 4 BIT MODE- ms nybble first


;    R/W- READ HIGH/WRITE LOW- tied low (no polling)





;**


; display ini


DINI	BCF	RBI,RS	       ;data with rs high


	MOVLW	0E0H	       ;preserve high three bits


	ANDWF	RBI,1	       ;and in back to image


	MOVLW	02	       ;write a 20 to disp (8 bits)


	IORWF	RBI,1	       ;get low to rbi


	MOVF	RBI,W	       ;get rb image to acc


	MOVWF	RB	       ;put on rb- data out


	NOP


	NOP


	BSF	RB,EN	       ;set enable bit high


	NOP


	NOP


	BCF	RB,EN	       ;set enable bit low


	NOP


;disp now in 4 bit mode


	MOVLW	20H	       ;4 bit mode


	CALL	TELL0	       ;send it


	MOVLW	0CH	       ;disp on


	CALL	TELL0	       ;send it


	MOVLW	06H	       ;set cursor


	CALL	TELL0	       ;send it


	MOVLW	01H	       ;clear disp


	CALL	TELL0	       ;send it


	RETLW	0	       ;done- dini


;************





;************


; say- send data in char (sets rs and does tell1)


SAY0	MOVWF	CHAR	       ;entry for data in w


SAY	BSF	RBI,RS	       ;data with rs high


	GOTO	TELL1	       ;jmp over clear- special entry


;************


; tell- generic write routine- write data in char


; three convenience entries


TELL0	MOVWF	CHAR	       ;entry for data in w


TELL	BCF	RBI,RS	       ;data with rs high


TELL1	MOVLW	02H	       ;load count- two passes


	MOVWF	COUNT	       ;store





LPTL	MOVLW	0E0H	       ;preserve high three bits


	ANDWF	RBI,1	       ;and in back to image


	SWAPF	CHAR,1	       ;get high nybble from char


	MOVF	CHAR,w	       ;get swapped to acc


	ANDLW	0FH	       ;mask low nybble


	IORWF	RBI,1	       ;get low to rbi





	MOVF	RBI,W	       ;get rb image to acc


	MOVWF	RB	       ;put on rb- data out


	BSF	RB,EN	       ;set enable bit high


	BCF	RB,EN	       ;set enable bit low





	DECFSZ	COUNT,1        ;two passes


	GOTO	LPTL	       ;loop twice- do low byte (swap happens)


	RETLW	0	       ;done


;


; ALL DONE








;  delay routine- general use for lcd inits, etc.





DLY	MOVLW	0ffH	       ;load in coarse


DLY2X	MOVWF	GP2	       ;put it in


DLLP2	CLRF	GP1	       ;use gp1 to count off ticks


DLLP1	DECFSZ	GP1,1	       ;dec gp1 value til 10 ms


	GOTO	DLLP1	       ;just loop


	DECFSZ	GP2,1	       ;dec gp2 value


	GOTO	DLLP2	       ;do another 10 ms til gp2=0


	RETLW	0	       ;return





;***************************************************************


; MAIN ROUTINE- DEBUG





START	CLRW		       ;clear ports preload


	MOVWF	RB	       ;port b low- all


	MOVWF	RBI	       ;copy image


	TRIS	RB	       ;all of b outputs





; port a-	  ra3 out dout, ra2 in din, ra1 out sclk, ra0 out cs,


;  pup defaults   0		0	    0		  1





	MOVLW	1H	       ;init ra


	MOVWF	RA	       ;mov to ra


	MOVLW	04H	       ;just bit 2 of ra is an input- din


	TRIS	RA	       ;set port ra ddr





	MOVLW	04H	       ;no wdt- rtcc from clock





	OPTION		       ;put in option register





	CALL	DLY


	CALL	DINI	       ;init display


	CALL	DLY





; debug- display a 0 on the display to start


	CLRW		       ;


	MOVWF	ACCbLO	       ;stash low


	MOVWF	ACCbHI	       ;stash hi


	CALL	B2_BCD	       ;convert to bcd


	CALL	PRF0	       ;display


	MOVLW	80H	       ;reset cursor


	CALL	TELL0


	CALL	DLY	       ;wait for finish





;main part of rouinte


;  sequence of events


;  clr rtcc-


;  turn on pos phase


;  wait 10 ms- rtcc still runs


;  configure and run a/d for pos phase


;  wait rtcc at 12 ms (even number)


;  turn on neg phase


;  wait 10 ms- rtcc still runs


;  configure and run a/d for neg phase


;  turn off both phases


;  do some arithmetic


;  display


;  repeat as necessary


;


; DEBUG


TUT	CLR	RTCC	       ;clr the rtcc counter


	BSF	RBI,SEN1       ;phase 1 (sen2 is low)


	MOVF	RBI,W	       ;get rb image to acc


	MOVWF	RB	       ;put on rb- data out





TUT1	BTFSC	RTCC,6	       ;wait for high on bit 6 of rtcc


	GOTO	TIMAD1	       ;tim for a/d


	GOTO	TUT1	       ;hang in loop





TIMAD1	MOVLW	082h	       ;DE/BIP


	CALL	CAD	       ;configure and start a/d


	NOP


	NOP


	NOP


	MOVLW	ACCaHI	       ;point to acc high byte- a


	CALL	GAD	       ;get a/d value





TUT2	BTFSC	RTCC,7	       ;wait for high on bit 7 of rtcc


	GOTO	TUT3	       ;tim for swap


	GOTO	TUT2	       ;hang in loop





TUT3	CLR	RTCC	       ;clr the rtcc counter


	BCF	RBI,SEN1       ;phase 2 (sen2 is low)


	BSF	RBI,SEN2       ;opposite phase


	MOVF	RBI,W	       ;get rb image to acc


	MOVWF	RB	       ;put on rb- data out





TUT4	BTFSC	RTCC,6	       ;wait for high on bit 6 of rtcc


	GOTO	TIMAD2	       ;tim for a/d


	GOTO	TUT4	       ;hang in loop





;a little house keeping while waiting


TIMAD2	MOVLW	0C2h	       ;DE/BIP


	CALL	CAD	       ;configure and start a/d


	NOP


	NOP


	NOP


	MOVLW	ACCbHI	       ;point to acc high byte- b


	CALL	GAD	       ;get a/d value





TUT5	BTFSC	RTCC,7	       ;wait for high on bit 7 of rtcc


	GOTO	TUT6	       ;tim for swap


	GOTO	TUT5	       ;hang in loop





TUT6	BCF	RBI,SEN1       ;all low


	BCF	RBI,SEN2       ;done


	MOVF	RBI,W	       ;get rb image to acc


	MOVWF	RB	       ;put on rb- data out





;ARITHMETIC SECTION


; measurements now done and values stowed


; first in acca, second in acc b


; since sign reversed between measurments


; offset is cancelled


; add (average) values together- 2's comp math


; do ascii minus and plus sign stuff





	MOVF	ACCALO,0       ;get first to w


	ADDWF	ACCBLO	       ;add (no carry)


	BTFSC	STATUS,CY      ;ck carry


	INCF	ACCbHI	       ;inc hi if cy


	MOVF	ACCaHI,W       ;add hi bytes


	ADDWF	ACCbHI	       ;add





;two's comp sum in accb- note doubled


;


	MOVLW	2BH	       ;preload a plus sign


	BTFSS	ACCBHI,4       ;check 13th bit


	GOTO	DSPLUS	       ;do positive





	COMF	ACCBLO	       ; negate ACCB


	INCF	ACCBLO


	BTFSC	STATUS,ZR


	DECF	ACCBHI


	COMF	ACCBHI





	MOVLW	0FH


	ANDWF	ACCBHI


	MOVLW	2DH	       ;entry- neg








DSPLUS	CALL	SAY0	       ;print it





;clamp values at +-F00h/ -0 AND 4095 COMPLEXITY


	INCF	ACCBHI


	BTFSS	ACCBHI,4       ;check


	GOTO	OK	       ;ok


	MOVLW	1	       ;ready for decf


	MOVWF	ACCBHI	       ;0


	CLRW		       ;zero


	MOVWF	ACCBLO	       ;make 0





;


OK	DECF	ACCBHI	       ;fixup


	CALL	B2_BCD	       ;convert to bcd


	CALL	PRF0	       ;display





	MOVLW	80H	       ;reset cursor


	CALL	TELL0


	CALL	DLY	       ;wait for finish





	GOTO	TUT








; END OF CODE


;********************************************************************


; VECTORS- start





	ORG	1FFH	       ;reset vector


	GOTO	START





	END

